

```

! Bond portfolio/cash matching problem in LINGO. Given:
cash needs in each of a series of future periods, what
collection of high quality bonds should we buy now,
so to meet these needs with certainty (the bonds do not default)?;
! We want to minimize initial cash needed;
! Keywords: @SEMIC, Bond portfolio, Cash flow matching, Covering, Defeasance,
LINGO, Maturity, Minimum buy, Semi-continuous ;
SETS:
    BOND :
        MATAT,    ! Maturity period;
        PRICE,    ! Price of bond;
        CAMNT,    ! Coupon payment each period;
        BUY,       ! Amount to buy;
        BUYMIN,   ! Minimum that must be bought if any;
        BUYMAX;   ! Maximum would ever buy;
    PERIOD:
        NEED,     ! Cash needed each period;
        SINVEST;  ! Short term investment each period;
ENDSETS

DATA:
    STRTE = 0.01;          ! Short term interest rate;
    BOND =   A       B       C;
    MATAT =   6,     13     15;  ! Years of maturity;
    PRICE = 0.980, 0.965, 0.99;  ! Bond purchase prices in 1000's;
    CAMNT = 0.060, 0.065, 0.064; ! Bond coupon payments in 1000's;
    BUYMIN=   5       5       9;
    BUYMAX=  140     140     130;
    NEED = 10, 11, 12, 14, 15, 17, 19, 20, 22, 24,
           26, 29, 31, 33, 15; ! Cash needs in 1000's for each period;
ENDDATA

SUBMODEL CHUZBNDS:
! Minimize the total investment required to generate
the stream of future cash needs;
    MIN = LUMP;
! First period is special, must cover first need,
plus buy bonds;
    LUMP = NEED( 1) + SINVEST( 1) +
        @SUM( BOND( b): PRICE( b) * BUY( b));
! For subsequent periods;
    @FOR( PERIOD( I)| I #GT# 1:
! Sources of cash = uses of cash, or
interest payments + maturities = needs to be covered;
        @SUM( BOND( b)| MATAT( b) #GE# I:
            CAMNT( b) * BUY( b)) +
            @SUM( BOND( b)| MATAT( b) #EQ# I:
                BUY( b)) +
            ( 1 + STRTE) * SINVEST( I - 1) =
            NEED( I) + SINVEST( I);
        );
! Can only buy integer bonds;
        @FOR( BOND( b): @GIN( BUY( b)));
! Enforce Min Buy quantities with semi-continuous constraint.
If we buy any, we must buy >= BUYMIN;
        @FOR( BOND( b): @SEMIC( BUYMIN( b), BUY( b), BUYMAX( b)));
ENDSUBMODEL

CALC:
! @GEN( CHUZBNDS); ! Generate display of model in scalar form;
! @SOLVE( CHUZBNDS); ! Solve the model;

! Write a little report;
@WRITE(' Initial Cash Needs in Period 1= ', @FORMAT( LUMP, '9.2f'), @NEWLINE( 2));
@WRITE(' Bond Maturity Price Coupon Units_to_buy', @NEWLINE(1));
@FOR( BOND( b):

```

```

@WRITE(' ', Bond(b), ' ', @FORMAT( MATAT( b), '3.0F')
      , ' ', @FORMAT( PRICE( b), '6.4F')
      , ' ', @FORMAT( CAMNT( b), '6.4F')
      , ' ', @FORMAT( BUY( b), '3.0F')
      , @NEWLINE(1));
);

```

```

ENDCALC

```